



# SAS® GLOBAL FORUM 2016



IMAGINE. CREATE. INNOVATE.

## **Using the REPORT Procedure to Export a Big Data set to an External XML**

#SASGF





# Using the REPORT Procedure to Export a Big Data set to an External XML

Guihong Chen

TCF Bank

## ABSTRACT

This paper uses PROC REPORT under the ODS environment to export a big data set into a colorful and customized xml file so that a non-SAS user can read the data easily in the xml file for useful information. DEFINE, COMPUTE, and COLUMN statements are used along with several options to present data in a more colorful way. It shows how to control the format of the selected columns and rows. The headings of columns can be more meaningful and pretty. It also talks about how to color any selected cells differently to bring attention to the audience. The data set used is called “Sashelp.snacks” provided by SAS in the Sashelp library. This data set provides daily snack food sales with 35,770 observations and 6 variables.

## STEP2: COLUMN CUSTOMIZATION (OUTPUT)

2. Suppress repetitious product printing      3.Format change to weekdate17.      4.Spanning column headers      5.More informative labels and split column headers

		Sales Details			
Product Name	Date of sale	Quantity Sold	Price	Holiday (1=yes)	Advertised (1=yes)
Baked potato chips	Tue, Aug 31, 2004	6	1.99	0	0
	Wed, Sep 1, 2004	1	1.99	1	0
	Thu, Sep 2, 2004	0	1.99	1	0
	Fri, Sep 3, 2004	3	1.99	1	0
	Sat, Sep 4, 2004	0	1.99	1	0

1.Change column order

## STEP1: EXPORTING DETAIL REPORT TO EXCEL

```
1 ods tagsets.excelxp file="C:\snacks.xml" style=normal
2 options(frozen_headers='yes' autofilter='all');
3 ods results off;
4 ods tagsets.excelxp options(sheet_name="Bread sticks");
5 proc print data=sashelp.SNACKS(where=(product="Bread sticks"));run;
6
7 ods tagsets.excelxp options(sheet_name="Cheese puffs");
8 proc print data=sashelp.SNACKS(where=(product="Cheese puffs"));run;
9
10 ods results on;
```

in two "tabs"

## STEP2: COLUMN CUSTOMIZATION (CODE)

```
1 proc report data=sashelp.SNACKS (where=(date>"30Aug2004"d) obs=5) split='#';
2 label price="Price";
3 column Product date ("Sales Details" QtySold Price Holiday Advertised );
4 define Product/order "Product#Name";
5 define Holiday/display "Holiday#(1=yes)";
6 define Advertised/display "Advertised#(1=yes)";
7 define date/display "Date# of sale" format=WEEKDATE17.;
8 define QtySold/display "Quantity#Sold";
9 run;
```

- Change the column order (Line 3) and Suppress repetitious column printing (Line 4)
- Change the formats (Line 7) and Specify a header to span multiple columns (Line 3)
- Change column labels (Line 2 and DEFINE statements) and Split the column headers

# Using the REPORT Procedure to Export a Big Data set to an External XML

Guihong Chen

TCF Bank

## STEP3: ROW CUSTOMIZATION (OUTPUT)

Add traffic lighting for positive sales

Date of sale	Sales Details			
	Quantity Sold	Price	Holiday (1=yes)	Advertised (1=yes)
Tue, Aug 31, 2004	6	1.99	0	0
Wed, Sep 1, 2004	1	1.99	1	0
Thu, Sep 2, 2004	0	1.99	1	0
Fri, Sep 3, 2004	3	1.99	1	0
Sat, Sep 4, 2004	0	1.99	1	0
Sun, Sep 5, 2004	0	1.99	1	0
Mon, Sep 6, 2004	0	1.99	1	0
Tue, Sep 7, 2004	0	1.99	1	0
Wed, Sep 8, 2004	1	1.99	0	0
Thu, Sep 9, 2004	3	1.99	0	0
Fri, Sep 10, 2004	3	1.99	0	0
Sat, Sep 11, 2004	0	1.99	0	0
Sun, Sep 12, 2004	8	1.99	0	0
Mon, Sep 13, 2004	0	1.99	0	0
Tue, Sep 14, 2004	6	1.99	0	0
Wed, Sep 15, 2004	0	1.99	0	0
Thu, Sep 16, 2004	0	1.99	0	0
Fri, Sep 17, 2004	7	1.49	0	1
Sat, Sep 18, 2004	16	1.49	0	1
Sun, Sep 19, 2004	5	1.49	0	1

Positive sales: yellow

Holiday sales: green

Advertised sales: red

## STEP3: ROW CUSTOMIZATION (CODE)

```
1 proc report data=sashelp.SNACKS (where=(date>"30Aug2004"d) obs=20)
2   split='#';label price="Price";
3   column  Product date ("Sales Details" QtySold Price Holiday Advertised);
4   define Product/order "Product#Name" noprint;
5   define Holiday/display "Holiday#(1=yes)";
6   define Advertised/display "Advertised#(1=yes)";
7   define date/display "Date# of sale" format=WEEKDATE17.;
8   define QtySold/"Quantity#Sold";
9   compute QtySold;if QtySold.sum>0 then
10  call define (_row_,'style','style={background=yellow}');endcomp;
11  compute Holiday; if QtySold.sum>0 and holiday=1 then
12  call define (_col_,'style','style={background=green font_weight=bold}');endcomp;
13  compute Advertised; if QtySold.sum>0 and Advertised=1 then
14  call define (_col_,'style','style={background=red font_style=italic}');endcomp;
15  run;
```

1. The first element of CALL DEFINE statement could be “\_row\_”, “\_col\_” or “\_c4\_”
2. The second element of CALL DEFINE statement could be “format”, “style” or “url”
3. Numerical variable with Analysis usage must use compound names

# Using the REPORT Procedure to Export a Big Data set to an External XML

Guihong Chen

TCF Bank

## STEP4: DISPLAY NEW VARIABLES (OUTPUT)

Product		Sales Details					
Product Name	Short Name	Date of sale	Quantity Sold	Price	Sales	Holiday (1=yes)	Advertised (1=yes)
Baked potato chips	Bpc	Tue, Aug 31, 2004	6	1.99	\$11.94	0	0
		Wed, Sep 1, 2004	1	1.99	\$1.99	1	0
		Thu, Sep 2, 2004	0	1.99	\$0.00	1	0
		Fri, Sep 3, 2004	3	1.99	\$5.97	1	0
		Sat, Sep 4, 2004	0	1.99	\$0.00	1	0

New Text Variable

New Numerical Variable

## STEP4: DISPLAY NEW VARIABLES (CODE)

```
1 proc report data=sashelp.SNACKS (where=(date>"30Aug2004"d) obs=5)
2   split='#';label price="Price";
3   column ("Product" Product Short_Name) date
4   ("Sales Details" QtySold Price Sales Holiday Advertised);
5   define Product/order "Product#Name";
6   define Holiday/display "Holiday#(1=yes)";
7   define Advertised/display "Advertised#(1=yes)";
8   define date/display "Date# of sale" format=WEEKDATE17.;
9   define QtySold/ "Quantity#Sold";
10
11   define sales/computed f=dollar20.2;
12   compute sales;sales=qtySold.sum*price.sum;endcomp;
13   define short_name/computed 'Short#Name' ;
14   compute short_name/character; short_name=substr(scan(product,1),1,1)||
15   substr(scan(product,2),1,1)||substr(scan(product,3),1,1);endcomp;
16 run;
```

1. Put the computed variable to the right of input variables
2. Numerical variable with Analysis usage must use compound names
3. Character variable must be computed with CHARACTER OR LENGTH option



# Using the REPORT Procedure to Export a Big Data set to an External XML

Guihong Chen

TCF Bank

## STEP5: ADD SUMMARY INFORMATION (OUTPUT)

Add summary information for all numerical variables.

Product Name	Date of sale	Sales Details				
		Quantity Sold	Price	Sales	Holiday (1=yes)	Advertised (1=yes)
Baked potato chips	Tue, Jan 1, 2002	0	1.99	\$0.00	0	0
	Wed, Jan 2, 2002	0	1.99	\$0.00	0	0
	Thu, Jan 3, 2002	0	1.99	\$0.00	0	0
	Fri, Jan 4, 2002	0	1.99	\$0.00	0	0
Baked potato chips		0	1.99	\$0.00		
Barbeque pork rinds	Tue, Jan 1, 2002	3	1.49	\$4.47	0	0
	Wed, Jan 2, 2002	11	1.49	\$16.39	0	0
	Thu, Jan 3, 2002	1	1.49	\$1.49	0	0
	Fri, Jan 4, 2002	1	1.49	\$1.49	0	0
Barbeque pork rinds		16	1.49	\$23.84		

## STEP5: ADD SUMMARY INFORMATION (CODE)

```
1 proc report data=sashelp.SNACKS (where=(date<"05Jan2002"d)) split='#';label price="Price";
2 column Product date ("Sales Details" QtySold Price Sales Holiday Advertised);
3 define Product/order "Product#Name";
4 define Holiday/display "Holiday#(1=yes)";
5 define Advertised/display "Advertised#(1=yes)";
6 define date/display "Date# of sale" format=WEEKDATE17.;
7 define QtySold/ "Quantity#Sold";
8 define price/mean;
9
10 define sales/computed f=dollar20.2;compute sales;sales=qtysold.sum*price.mean;endcomp;
11 break after product/summarize;
12 run;
```

1. You can only BREAK on an ORDER or GROUP variable
2. Add SUMMARIZE option for the BREAK statement
3. SUM is the default statistics for a numerical variable

# Using the REPORT Procedure to Export a Big Data set to an External XML

Guihong Chen

TCF Bank

## STEP6: ADD CUSTOMIZED SUMMARY INFORMATION (OUTPUT)

Customize the summary information to display what interests us most

		Sales Details				
Product Name	Date of sale	Quantity Sold	Price	Sales	Holiday (1=yes)	Advertised (1=yes)
Baked potato chips has total sales of		\$0.00				
Baked potato chips	Tue, Jan 1, 2002	0	1.99	\$0.00	0	0
	Wed, Jan 2, 2002	0	1.99	\$0.00	0	0
	Thu, Jan 3, 2002	0	1.99	\$0.00	0	0
	Fri, Jan 4, 2002	0	1.99	\$0.00	0	0
Barbeque pork rinds has total sales of		\$23.84				
Barbeque pork rinds	Tue, Jan 1, 2002	3	1.49	\$4.47	0	0
	Wed, Jan 2, 2002	11	1.49	\$16.39	0	0
	Thu, Jan 3, 2002	1	1.49	\$1.49	0	0
	Fri, Jan 4, 2002	1	1.49	\$1.49	0	0

## STEP6: ADD CUSTOMIZED SUMMARY INFORMATION (CODE)

```
1 proc report data=sashelp.SNACKS(where=(date<"05Jan2002"d)) split='#';label price="Price";
2 column Product date ("Sales Details" QtySold Price Sales Holiday Advertised);
3 define Product/order "Product#Name" ;
4 define Holiday/display "Holiday#(1=yes)";
5 define Advertised/display "Advertised#(1=yes)";
6 define date/display "Date# of sale" format=WEEKDATE17.;
7 define QtySold/ "Quantity#Sold";
8 define price/mean;
9
10 define sales/computed f=dollar20.2;
11 compute sales;sales=qtysold.sum*price.mean;endcomp;
12 compute before product;line product $20. "has total sales of " sales dollar20.2;endcomp;
13 run;
```

1. LINE Statement is only valid in COMPUTE BEFOR or COMPUTE AFTER block
2. LINE statement is executed after the other statements in the block
3. Change the default statistics SUM to MEAN for Price

## CONCLUSIONS

To produce a detail report, you don't have the same flexibility with PROC SQL and PRINT as you do with PROC REPORT which provides a handful of options and statements users can leverage to control the appearance of the output of the data extensively.

## REFERENCES

Zender, Cynthia. 2009. SAS ® Report Writing 1: Using Procedures and ODS. Cary, NC: SAS Institute Inc.





# SAS<sup>®</sup> GLOBAL FORUM 2016

IMAGINE. CREATE. INNOVATE.

LAS VEGAS | APRIL 18-21

#SASGF